Differenciálegyenletek Mathematica - val

Krizsán Lívia, 2008

Késleltetett argumentumú differenciálegyenletek

Késleltett argumentumú differenciálegyenlet :

$$y'(x) = f(x, y(x), y(x-\tau)) \qquad (\tau > 0 \text{ konstans})$$

Példa : $y'(x) == x - 2\,y(x) + y(x-3);$

Lépések módszere

A $\varphi(x)$ függvényt <u>kezdeti függvény</u>nek nevezzük, ha :

$$\varphi(x) = y(x), \qquad x_0 - \tau \leq x \leq x_0$$

In[1]:=
```
φ[x_] := x + 3;
x0 = 0; τ = 3;
sh1 = Show[Plot[φ[x], {x, x0 - τ, x0}, AspectRatio → Automatic,
    PlotRange → {{x0 - τ - 0.2, x0 + 2 * τ + 0.2}, {-0.2, 5}}, PlotStyle -> {Hue[0.3]}],
  Graphics[{PointSize[Large], Red, Point[{-τ, φ[-τ]}]}],
  Graphics[{PointSize[Large], Red, Point[{x0, φ[x0]}]}]]
```

Out[3]=



Jelöljük (∗)−gal a következő egyenletrendszert:

$$y'(x) = f(x, y(x), y(x-\tau)) \qquad (\tau > 0 \text{ konstans})$$
$$\varphi(x) = y(x), \qquad x_0 - \tau \leq x \leq x_0$$

Lépések módszerével belátható,
hogy *a* (∗) probléma megoldása több k.é.p. megoldására vezetődik vissza.

Keressük (∗) megoldását az $[x_0, x_0 + (n+1)\tau]$ intervallumon!

Legyen $x \in [x_0, x_0 + \tau] \Rightarrow x - \tau \in [x_0 - \tau, x_0] \Rightarrow y(x - \tau) = \varphi(x - \tau)$

1. *K.é.p.*:

$$y'(x) = f(x, y(x), \varphi(x - \tau))$$
$$y(x_0) = \varphi(x_0)$$

Tfh. ∃! $y(x) := \varphi_1(x)$ megoldása 1. *K.é.p.*–nak $[x_0, x_0 + \tau]$ –n.

In[4]:=
```
φ1[x_] =
  y[x] /.
   NDSolve[{y'[x] == x - 2 * y[x] + φ[x - 3], y[x0] == φ[x0]}, y[x], {x, x0, x0 + τ}][[1]];
sh2 = Show[sh1, Plot[φ1[x], {x, x0, x0 + τ}, AspectRatio → Automatic,
   PlotRange → {{x0 - τ - 0.2, x0 + 2 * τ + 0.2}, {-0.2, 5}}, PlotStyle -> {Hue[0.9]}],
  Graphics[{PointSize[Large], Red, Point[{x0 + τ, φ1[x0 + τ]}]}]]
```

Out[5]=


Legyen $x \in [x_0 + \tau,\ x_0 + 2\tau] \Rightarrow x - \tau \in [x_0,\ x_0 + \tau] \Rightarrow y(x - \tau) = \varphi_1(x - \tau)$

2. *K.é.p.*:

$$y'(x) = f(x,\ y(x),\ \varphi_1(x - \tau))$$
$$y(x_0 + \tau) = \varphi_1(x_0 + \tau)$$

Tfh. ∃ ! $y(x) := \varphi_2(x)$ megoldása 2. *K.é.p.* – nak $[x_0 + \tau,\ x_0 + 2\tau]$ – n.

```
In[6]:= φ2[x_] =
          z[x] /. NDSolve[{z'[x] == x - 2 * z[x] + φ1[x - 3], z[x0 + τ] == φ1[x0 + τ]},
              z[x], {x, x0 + τ, x0 + 2 * τ}][[1]];
        sh3 = Show[sh2, Plot[φ2[x], {x, x0 + τ, x0 + 2 * τ}, AspectRatio → Automatic,
           PlotStyle -> {Hue[0.5]}, PlotRange → {{x0 - τ - 0.2, x0 + 2 * τ + 0.2}, {-0.2, 5}}],
          Graphics[{PointSize[Large], Red, Point[{x0 + 2 * τ, φ2[x0 + 2 * τ]}]}]]
```

Out[7]=



...

Legyen $x \in [x_0 + (n-1)\tau,\ x_0 + n\tau] \Rightarrow x - \tau \in [x_0 + (n-2)\tau,\ x_0 + (n-1)\tau,] \Rightarrow y(x-\tau) = \varphi_{n-1}(x-\tau)$

<u>*n.* *K*.é.*p.*</u>:

$$y'(x) = f(x,\ y(x),\ \varphi_{n-1}(x-\tau))$$
$$y(x_0 + (n-1)\tau) = \varphi_{n-1}(x_0 + (n-1)\tau)$$

Tfh. $\exists!\ y(x) := \varphi_n(x)$ megoldása *n. K*.é.*p.* – nak $[x_0 + (n-1)\tau,\ x_0 + n\tau]$ – n.

A lépések módszerével minden korlátos intervallumon megadható $(*)$–nak egy $y(x)$ megoldása, feltéve hogy egyértelműen léteznek a fellépő *k*.é.*p.*–k megoldásai az adott intervallumokon.

In[8]:=
```
Animate[For[i = 1; φ[x_] = x + 3; sh = Show[Plot[φ[x], {x, x0 - τ, x0},
    PlotStyle -> {Hue[0.1]}, PlotRange -> {{x0 - τ - 0.2, x0 + 10 * τ + 0.2}, {-0.2, 30}}],
   Graphics[{PointSize[Large], Red, Point[{x0 - τ, φ[x0 - τ]}]}],
   Graphics[{PointSize[Large], Red, Point[{x0, φ[x0]}]}]], i < n + 1, i++,
  φ[x_] = z[x] /. NDSolve[{z'[x] == x - 2 * z[x] + φ[x - 3], z[x0 + (i - 1) τ] == φ[x0 + (i - 1) τ]},
      z[x], {x, x0 + (i - 1) τ, x0 + i * τ}][[1]]; sh = Show[sh, Plot[φ[x],
     {x, x0 + (i - 1) τ, x0 + i * τ}, PlotRange -> {{x0 - τ - 0.2, x0 + 10 * τ + 0.2}, {-0.2, 30}},
     PlotStyle -> {Hue[i * 0.2 + 0.1]}], Graphics[
     {PointSize[Large], Red, Point[{x0 + i * τ, φ[x0 + i * τ]}]}]]]; Show[sh], {n, 0, 10, 1}]
```

Out[8]=



■ Megjegyzés

A lépések módszere működik az

$$y'(x) = f(x,\ y(x),\ y(x-\tau)) \qquad (\tau > 0 \text{ konstans})$$

esetben is.

Példa :

$$y'(x) = y(x) + y\left(x - \frac{1}{2}\right) y'\left(x - \frac{1}{2}\right) y(x), \qquad x \in \left(-\frac{1}{2}, \frac{1}{2}\right)$$

$$\textbf{Tudjuk: } y(x) = 1, \quad -1 \le x \le -\frac{1}{2}$$

Megoldás :

In[9]:=
```
Clear[i, n, x0, τ, φ, r, sh, x, z]
```

In[10]:=
```
n = 2; x0 = -1 / 2; τ = 1 / 2; r = 8;
```

In[11]:=
```
For[i = 1; φ[x_] = 1; sh = Show[Plot[φ[x], {x, x0 - τ, x0},
    PlotStyle -> {Hue[0.1]}, PlotRange -> {{x0 - τ - 0.2, x0 + n * τ + 0.2}, {-0.2, r}}],
   Graphics[{PointSize[Large], Red, Point[{x0 - τ, φ[x0 - τ]}]}],
   Graphics[{PointSize[Large], Red, Point[{x0, φ[x0]}]}]], i < n + 1,
 i++, φ[x_] = z[x] /. NDSolve[{z'[x] == z[x] + φ[x - 1 / 2] φ'[x - 1 / 2] z[x],
      z[x0 + (i - 1) τ] == φ[x0 + (i - 1) τ]}, z[x], {x, x0 + (i - 1) τ, x0 + i * τ}][[1]];
 sh = Show[sh, Plot[φ[x], {x, x0 + (i - 1) τ, x0 + i * τ}, PlotRange ->
     {{x0 - τ - 0.2, x0 + n * τ + 0.2}, {-0.2, r}}, PlotStyle -> {Hue[i * 0.2 + 0.1]}],
   Graphics[{PointSize[Large], Red, Point[{x0 + i * τ, φ[x0 + i * τ]}]}]]]
```

In[12]:=
```
Show[sh]
```

Out[12]=



■ Differenciálegyenletek megoldásainak közelítése Euler - módszerrel

Adott az

$$x'(t) = f(t, x(t))$$
$$x(t_0) = x_0$$

k.é.p.

$$x(t) = x_0 + \int_{t_0}^{t} f(s, x(s))\, ds$$

megoldással.

Euler-módszer

<u>Ötlet</u> : Egy adott $[a, b]$ intervallumon szakaszonként közelítsük a megoldást lineáris függvénnyel!

Legyen ∇t rögzített.

$$t_0 := a \qquad x_0(t) := x_0$$
$$t_1 := t_0 + \nabla t \qquad x_1(t) := x_0 + \nabla t\, f(t_0,\ x_0)$$
$$t_2 := t_1 + \nabla t \qquad x_2(t) := x_1 + \nabla t\, f(t_1,\ x_1)$$
...
$$t_k := t_{k-1} + \nabla t \qquad x_k(t) := x_{k-1} + \nabla t\, f(t_{k-1},\ x_{k-1})$$

<u>Példa1</u> :

$$x'(t) = \cos(t)$$
$$x(0) = 0$$

```
In[13]:= Clear[mo, x, t, a, b, f, x0, plt, t0, X, dt, i]

In[14]:= a = 0; b = 2 Pi; f[x_, t_] = Cos[t]; x0 = 0; t0 = 0;

In[15]:= mo[t_] = x[t] /. NDSolve[{x'[t] == f[x, t], x[t0] == x0}, x[t], {t, a, b}][[1]];

In[16]:= plt = Plot[mo[t], {t, a, b}, AspectRatio → 1 / 2, PlotStyle → {Thickness[0.005], Hue[0]}]
```

Out[16]=



In[17]:=
```
Manipulate[X[n_] := X[n - 1] + dt * (f[x, t] /. x[t] → X[n - 1]) /. t → a + (n - 1) dt;
 X[0] := x0; Show[Graphics[Line[Table[{a + i * dt, X[i]}, {i, 0, b / dt}]]],
  plt, Axes → True, AspectRatio → 1 / 2], {{dt, Pi / 2, "lépésköz"}, Pi / 2, Pi / 8}]
```

Out[17]=



Példa2 :

$$x'(t) = -x(t)$$
$$x(0) = 1$$

In[18]:=
```
Clear[mo, x, t, a, b, f, x0, plt, t0, X, dt]
```

In[19]:=
```
a = 0; b = 2; f[x_, t_] = -x[t]; x0 = 1; t0 = 0;
```

In[20]:=
```
mo[t_] = x[t] /. NDSolve[{x'[t] == f[x, t], x[t0] == x0}, x[t], {t, a, b}][[1]];
```

In[21]:=
```
plt = Plot[mo[t], {t, a, b}, AspectRatio → Automatic,
  AxesOrigin → {0, 0}, PlotStyle → {Thickness[0.005], Hue[0]}]
```

Out[21]=



In[22]:=
```
Manipulate[X[n_] := X[n - 1] + dt * (f[x, t] /. x[t] → X[n - 1]) /. t → a + (n - 1) dt;
 X[0] := x0; Show[Graphics[Line[Table[{a + i * dt, X[i]}, {i, 0, b / dt}]]], plt, Axes → True,
  AspectRatio → Automatic, AxesOrigin → {0, 0}], {{dt, 1, "lépésköz"}, 1, 1 / 8}]
```

Out[22]=